

Lab 2 寄存器堆与存储器及其应用

姓名:傅申 学号: PB20000051 实验日期: 2022-3-15

1 实验题目

寄存器堆与存储器及其应用

2 实验目的

- 掌握寄存器堆 (Register File) 和存储器的功能, 时序及其应用
- 熟练掌握数据通路和控制器的设计和描述方法

3 实验平台

- Xilinx Vivado v2019.1
- Microsoft Visual Studio Code
- FPGAOOL

4 实验过程


4.1 寄存器堆

寄存器堆的 Verilog 代码如下

```
1 // 32 * WIDTH Register File
2 module register_file #(
3     parameter WIDTH = 32
4 ) (
5     input  clk,                // Clock (posedge)
6     input  [4:0] ra0,          // Read address 0
7     input  [4:0] ra1,          // Read address 1
8     input  [4:0] wa,           // Write address
9     input  we,                 // Write enable
10    input  [WIDTH-1:0] wd,      // Write data
11    output [WIDTH-1:0] rd0,     // Read data 0
12    output [WIDTH-1:0] rd1     // Read data 1
13 );
14    reg [WIDTH-1:0] regfile [0:31];
15    assign rd0 = regfile[ra0];
16    assign rd1 = regfile[ra1];
17    always @(posedge clk) begin
18        if (we) regfile[wa] <= wd;
19    end
20 endmodule //register_file
```

4.2 RAM 存储器

4.2.1 分布式 16×8 位单端口 RAM



其中 `coe` 文件如下

```
1 | memory_initialization_radix=16;  
2 | memory_initialization_vector=0F 1E 2D 3C 4B 5A 69 78  
3 |                               87 96 A5 B4 C3 D2 E1 F0;
```

4.2.2 块式 16×8 位单端口 RAM

Re-customize IP

Block Memory Generator (8.4)

[Documentation](#) [IP Location](#) [Switch to Defaults](#)

IP SymbolPower Estimation

☐ Show disabled ports

BRAM_PORTA

▶ addra[3:0]

▶ clka

▶ dina[7:0]

▶ douta[7:0]

▶ ena

▶ wea[0:0]

Component Nameblock_ram

BasicPort A OptionsOther OptionsSummary

Interface TypeNative

Memory TypeSingle Port RAM

☐ Generate address interface with 32 bits

☐ Common Clock

ECC Options

ECC TypeNo ECC

☐ Error Injection PinsSingle Bit Error Injection

Write Enable

☐ Byte Write Enable

Byte Size (bits)9

Algorithm Options


Defines the algorithm used to concatenate the block RAM primitives.
Refer datasheet for more information.

AlgorithmMinimum Area

Primitive8kx2

OK

Cancel




其中 **coe** 文件如下

```


1 | memory_initialization_radix=16;
2 | memory_initialization_vector=0F 1E 2D 3C 4B 5A 69 78
3 |                               87 96 A5 B4 C3 D2 E1 F0;
```

4.3 利用寄存器堆实现 FIFO 队列

FIFO 队列的输入输出如下图



它的数据通路如下, 其中 **enq** 与 **deq** 信号都由 **SEDG** 模块两级同步并取边缘后输入到 **LCU** 模块.



在 Vivado 项目中, 设计文件层次如下

```
fifo (FIFO.v)
├── RF:      reg_file      (RegisterFile.v)
├── SEDG_enq: sig_edge     (SignalEdge.v)
├── SEDG_deq: sig_edge     (SignalEdge.v)
├── LCU:     list_control_unit (ListControlUnit.v)
└── SDU:     segplay_unit    (SegDisplayUnit.v)
```


4.3.1 寄存器堆模块

该模块实现了一个 8×16 位寄存器堆.

```
1  // A 8*16 register file
2  module reg_file (
3      input  clk,          // clock (posedge)
4      input  [2:0] ra0,    // read address 0
5      input  [2:0] ra1,    // read address 1
6      input  we,           // write enable
7      input  [2:0] wa,     // write address
8      input  [3:0] wd,     // write data
9      output [3:0] rd0,    // read data 0
10     output [3:0] rd1     // read data 1
11 );
12     reg [3:0] regfile [0:7];
13     assign rd0 = regfile[ra0];
14     assign rd1 = regfile[ra1];
15     always @(posedge clk) begin
16         if (we) regfile[wa] = wd;
17     end
18 endmodule //reg_file
```

4.3.2 取边缘模块

该模块按照下面的电路对信号进行两级同步并取边缘.



```

1  module sig_edge(
2      input  clk, // clock (100 MHz, posedge)
3      input  a,   // input signal
4      output s,   // synchronized signal
5      output p    // signal edge
6  );
7      reg a_reg_0;
8      reg a_reg_1;
9      reg a_reg_2;
10     always @(posedge clk) begin
11         a_reg_0 <= a;
12         a_reg_1 <= a_reg_0;
13         a_reg_2 <= a_reg_1;
14     end
15     assign s = a_reg_1;
16     assign p = a_reg_1 & ~a_reg_2;
17 endmodule

```

4.3.3 列控制单元模块

该模块处理输入信号, 对寄存器堆进行写入, 并向显示单元发送 **valid** 信号. 由于 **enq** 与 **deq** 在经过取边缘后每次有效的时间都是一个周期, 并且进入三个状态的条件与当前状态, 所以这里没有用到状态机.

在该模块中, 我定义了两个寄存器变量: 头指针 **head** 与尾指针 **tail**, 两者都指向下一次入/出队的数据下标, 即初始值均为 0. 在入队后, 尾指针 **tail** 自增, 在出队后, 头指针 **head** 自增. 对于判断队列是否为空 (满), 这里只需要判断 **valid** 的每一位是否都是 0 (1). 模块的具体代码如下

```

1  module list_control_unit(
2      input          clk,           // clock (100 MHz, posedge)
3      input          rst,           // synchronous reset (active high)
4      input [3:0]    in,
5      input          enq,           // enqueue edge
6      input          deq,           // dequeue edge
7      input [3:0]    rd,           // read data
8      output         full,
9      output         emp,           // empty
10     output reg [3:0] out,           // dequeued data
11     output [2:0]    ra,           // read address
12     output          we,           // write enable
13     output [2:0]    wa,           // write address
14     output [3:0]    wd,           // write data
15     output reg [7:0] valid

```

```

16 );
17     reg [2:0] head; // pointer to head
18     reg [2:0] tail; // pointer to tail
19
20     assign full = &valid;
21     assign emp  = ~(|valid);
22
23     assign ra = head;
24     assign we = enq & ~full & ~rst;
25     assign wa = tail;
26     assign wd = in;
27
28     always @(posedge clk) begin
29         if (rst) begin
30             valid <= 8'h00;
31             head  <= 3'h0;
32             tail  <= 3'h0;
33             out   <= 3'h0;
34         end
35         else if(enq & ~full) begin
36             valid[tail] <= 1'b1;
37             tail        <= tail + 3'h1;
38         end
39         else if(deq & ~emp) begin
40             valid[head] <= 1'b0;
41             head        <= head + 3'h1;
42             out         <= rd;
43         end
44     end
45 endmodule

```

4.3.4 数码管显示单元模块

在显示单元中, 输入的时钟信号是 100MHz 的, 对于数码管来说频率过快, 因此这里对其降频到 400Hz.

我使用了一个 18 位的模 250000 计数器, 在每个 100MHz 的时钟上升沿进行计数, 并在进位 (计数器值大于等于 249999) 时对输出的地址 (输出到寄存器堆的 **ra0**) 进行加一, 实现对寄存器堆 400Hz 的扫描. 同时, 在计数器值为 1 时会输出一个脉冲, 用于控制数码管信号的输出.

若队列为空, 则显示单元会输出信号使得最低位数码管显示 0, 否则会使得有效位的数码管上显示该位的值.

具体的模块代码如下

```

1 module segplay_unit(
2     input          clk_100mhz,
3     input          [3:0] data,
4     input          [7:0] valid,
5     output reg     [2:0] addr,
6     output         [2:0] segplay_an,
7     output         [3:0] segplay_data
8 );

```

```

9      // Counter: slow the clock down to 400Hz
10     wire clk_400hz;
11     reg [17:0] clk_cnt;
12     assign clk_400hz = ~(clk_cnt);    // clk_400hz = (clk_cnt == 0)
13     always @(posedge clk_100mhz) begin
14         if (clk_cnt >= 18'h3D08F) begin // clk_cnt >= 249999
15             clk_cnt <= 18'h00000;
16             addr <= addr + 3'b001;
17         end else
18             clk_cnt <= clk_cnt + 18'h00001;
19     end
20     // Generate segplay output
21     reg [2:0] segplay_an_reg;
22     reg [3:0] segplay_data_reg;
23     always @(posedge clk_100mhz) begin
24         if (clk_400hz && valid[addr]) begin
25             segplay_an_reg <= addr;
26             segplay_data_reg <= data;
27         end
28     end
29     assign segplay_data = (|valid) ? segplay_data_reg : 4'h0;
30     assign segplay_an = (|valid) ? segplay_an_reg : 3'h0;
31 endmodule

```

4.3.5 顶层模块 FIFO

顶层模块主要做连线任务, 这里不多赘述.

```

1  module fifo(
2      input          clk,      // clock (100 MHz, posedge)
3      input          rst,      // synchronous reset (active high)
4      input          enq,      // enqueue (active high)
5      input [3:0]    in,       // enqueue data
6      input          deq,      // dequeue (active high)
7      output [3:0]   out,      // dequeue data
8      output         full,     // queue full
9      output         emp,      // queue empty
10     output [2:0]    an,       // segment display selection
11     output [3:0]    seg       // segment display data
12 );
13     // wires
14     wire enq_edge;
15     wire deq_edge;
16     wire we;
17     wire [2:0] ra0, ra1, wa;
18     wire [3:0] rd0, rd1, wd;
19     wire [7:0] valid;
20     // datapath
21     reg_file RF(
22         .clk(clk),
23         .ra0(ra0),
24         .ra1(ra1),
25         .we (we),
26         .wa (wa),
27         .wd (wd),

```



```


28         .rd0(rd0),
29         .rd1(rd1)
30     );
31
32     sig_edge SEDG_enq(
33         .clk(clk),
34         .a  (enq),
35         .p  (enq_edge)
36     );
37     sig_edge SEDG_deq(
38         .clk(clk),
39         .a  (deq),
40         .p  (deq_edge)
41     );
42
43     list_control_unit LCU(
44         .clk  (clk),
45         .rst  (rst),
46         .in   (in),
47         .enq  (enq_edge),
48         .deq  (deq_edge),
49         .rd   (rd0),
50         .full (full),
51         .emp  (emp),
52         .out  (out),
53         .ra   (ra0),
54         .we   (we),
55         .wa   (wa),
56         .wd   (wd),
57         .valid(valid)
58     );
59
60     segplay_unit SDU(
61         .clk_100mhz  (clk),
62         .data         (rd1),
63         .valid        (valid),
64         .addr         (ra1),
65         .segplay_an   (an),
66         .segplay_data(seg)
67     );
68 endmodule

```

5 实验结果


5.1 寄存器堆

针对寄存器堆的[仿真文件](#), 仿真结果如下:



5.2 RAM 存储器

针对RAM存储器的[仿真文件](#), 仿真结果如下:




可以看到块存储器的输出要比输入慢, 并且 `ena` 信号可以控制块存储器的输出.

5.3 FIFO 队列

5.3.1 仿真

针对FIFO队列的[仿真文件](#), 仿真结果如下:




其中数码管显示信号没有进行仿真.

5.3.2 下载测试

因为在检查中已经演示过了, 这里只展示几个特殊的场景. 或者在[这里](#)查看视频.


- 队空/初始状态



- 队满



- 循环队列展示



6 心得体会

实验难度适中.

7 附录: 仿真文件

7.1 寄存器堆仿真文件

寄存器堆[仿真结果](#)

```
1  `timescale 1ns / 1ps
2  module testbench();
3      parameter clk_sep    = 1;
4      parameter time_sep   = 10;
5      parameter width      = 32;
6      reg      clk;
7      reg      [4:0] ra0;
8      reg      [4:0] ra1;
9      reg      [4:0] wa;
10     reg      we;
11     reg      [width-1:0] wd;
12     wire     [width-1:0] rd0;
13     wire     [width-1:0] rd1;
14     register_file regfile(
15         .clk(clk),
16         .ra0(ra0),
17         .ra1(ra1),
18         .wa(wa),
19         .we(we),
20         .wd(wd),
21         .rd0(rd0),
22         .rd1(rd1)
23     );
24     initial begin
25         clk = 0;
26         ra0 = 5'h03;
27         ra1 = 5'h12;
28         forever #clk_sep clk = ~clk;
29     end
30     initial begin
31         we = 1'b0;
32         wa = 5'h03;
33         wd = 32'h12345678;
34         #time_sep
35         we = 1'b1;
36         #time_sep
37         wa = 5'h12;
38         wd = 32'h87654321;
39         #time_sep
40         we = 1'b0;
41         wd = 32'habcdef01;
42         #time_sep
43         we = 1'b1;
44         #time_sep
45         we = 1'b0;
46         wa = 5'h03;
47         #time_sep
48         we = 1'b1;
```

```

49         #time_sep
50         $finish;
51     end
52 endmodule

```

寄存器堆[仿真结果](#)

7.2 RAM存储器仿真文件

RAM存储器[仿真结果](#)

```

1  module tb();
2      // shared input signals
3      reg      clk;
4      reg [3:0] addr;
5      reg [7:0] in;
6      reg      we;
7      initial begin
8          clk <= 1'b0;
9          forever
10             #1 clk <= ~clk;
11     end
12     initial begin
13         addr <= 4'h0;
14         in  <= 8'h00;
15         we  <= 1'b0;
16         #10 addr <= 4'h1;
17         #10 addr <= 4'h2;
18         #10 addr <= 4'h3;
19         #10 addr <= 4'h4;
20         #10 addr <= 4'h5;
21         #10 addr <= 4'h6;
22         #10 addr <= 4'h7;
23         #10 addr <= 4'h8;
24         #10 addr <= 4'h9;
25         #10 addr <= 4'hA;
26         #10 addr <= 4'hB;
27         #10 addr <= 4'hC;
28         #10 addr <= 4'hD;
29         #10 addr <= 4'hE;
30         #10 addr <= 4'hF;
31         #10 addr <= 4'h0;
32         in  <= 8'h00;
33         we  <= 1'b1;
34         #10 addr <= 4'h1;
35         in  <= 8'h11;
36         #10 addr <= 4'h2;
37         in  <= 8'h22;
38         #10 addr <= 4'h3;
39         in  <= 8'h33;
40         #10 addr <= 4'h4;
41         in  <= 8'h44;
42         #10 addr <= 4'h5;
43         in  <= 8'h55;
44         #10 addr <= 4'h6;

```

```

45         in    <= 8'h66;
46         #10 addr <= 4'h7;
47         in    <= 8'h77;
48         #10 addr <= 4'h8;
49         in    <= 8'h88;
50         #10 addr <= 4'h9;
51         in    <= 8'h99;
52         #10 addr <= 4'hA;
53         in    <= 8'hAA;
54         #10 addr <= 4'hB;
55         in    <= 8'hBB;
56         #10 addr <= 4'hC;
57         in    <= 8'hCC;
58         #10 addr <= 4'hD;
59         in    <= 8'hDD;
60         #10 addr <= 4'hE;
61         in    <= 8'hEE;
62         #10 addr <= 4'hF;
63         in    <= 8'hFF;
64         #10 addr <= 4'h0;
65         we    <= 1'b0;
66         #10 addr <= 4'h1;
67         #10 addr <= 4'h2;
68         #10 addr <= 4'h3;
69         #10 addr <= 4'h4;
70         #10 addr <= 4'h5;
71         #10 addr <= 4'h6;
72         #10 addr <= 4'h7;
73         #10 addr <= 4'h8;
74         #10 addr <= 4'h9;
75         #10 addr <= 4'hA;
76         #10 addr <= 4'hB;
77         #10 addr <= 4'hC;
78         #10 addr <= 4'hD;
79         #10 addr <= 4'hE;
80         #10 addr <= 4'hF;
81         #10 $finish;
82     end
83     // block memory
84     reg          ena;
85     wire [7:0] out_block;
86     initial begin
87         ena <= 1'b1;
88         #330 ena <= 1'b0;
89         forever
90             #5 ena <= ~ena;
91     end
92     block_ram          test_block(
93         .clka(clk),
94         .addra(addr),
95         .dina(in),
96         .douta(out_block),
97         .ena(ena),
98         .wea(we)
99     );
100     // distributed memory
101     wire [7:0] out_dist;
102     distributed_ram test_dist(

```



```

103         .clk(clk),
104         .a(addr),
105         .d(in),
106         .we(we),
107         .spo(out_dist)
108     );
109 endmodule

```

RAM 存储器[仿真结果](#)

7.3 FIFO队列仿真文件

FIFO队列[仿真结果](#)

```

1  `timescale 1ns / 1ps
2  module tb();
3      reg      clk;
4      reg      rst;
5      reg      enq;
6      reg      deq;
7      reg  [3:0] in;
8      wire  [3:0] out;
9      wire      full;
10     wire      emp;
11
12     fifo      test(
13         .clk(clk),
14         .rst(rst),
15         .enq(enq),
16         .deq(deq),
17         .in(in),
18         .out(out),
19         .full(full),
20         .emp(emp)
21     );
22
23     initial begin
24         clk <= 1'b0;
25         forever
26             #1 clk <= ~clk;
27     end
28
29     initial begin
30         rst <= 1'b1;
31         #5  rst <= 1'b0;
32     end
33
34     initial begin
35         enq <= 1'b0;
36         deq <= 1'b0;
37         in  <= 4'h0;
38         #20 enq <= 1'b1;    // 1st enqueue
39         in  <= 4'h1;
40         #20 enq <= 1'b0;
41         #20 enq <= 1'b1;    // 2nd enqueue

```

```

42     in  <= 4'h2;
43     #20 enq <= 1'b0;
44     #20 enq <= 1'b1;    // 3rd enqueue
45     in  <= 4'h3;
46     #20 enq <= 1'b0;
47     #20 enq <= 1'b1;    // 4th enqueue
48     in  <= 4'h4;
49     #20 enq <= 1'b0;
50     #20 enq <= 1'b1;    // 5th enqueue
51     in  <= 4'h5;
52     #20 enq <= 1'b0;
53     #20 enq <= 1'b1;    // 6th enqueue
54     in  <= 4'h6;
55     #20 enq <= 1'b0;
56     #20 enq <= 1'b1;    // 7th enqueue
57     in  <= 4'h7;
58     #20 enq <= 1'b0;
59     #20 enq <= 1'b1;    // 8th enqueue
60     in  <= 4'h8;
61     #20 enq <= 1'b0;
62     #20 enq <= 1'b1;    // 9th enqueue (invalid)
63     in  <= 4'h9;
64     #20 enq <= 1'b0;
65     #20 enq <= 1'b1;    // 10th enqueue (invalid)
66     in  <= 4'hA;
67     #20 enq <= 1'b0;
68     #20 deq <= 1'b1;    // 1st dequeue
69     #20 deq <= 1'b0;
70     #20 deq <= 1'b1;    // 2nd dequeue
71     #20 deq <= 1'b0;
72     #20 deq <= 1'b1;    // 3rd dequeue
73     #20 deq <= 1'b0;
74     #20 deq <= 1'b1;    // 4th dequeue
75     #20 deq <= 1'b0;
76     #20 deq <= 1'b1;    // 5th dequeue
77     #20 deq <= 1'b0;
78     #20 deq <= 1'b1;    // 6th dequeue
79     #20 deq <= 1'b0;
80     #20 deq <= 1'b1;    // 7th dequeue
81     #20 deq <= 1'b0;
82     #20 deq <= 1'b1;    // 8th dequeue
83     #20 deq <= 1'b0;
84     #20 deq <= 1'b1;    // 9th dequeue (invalid)
85     #20 deq <= 1'b0;
86     #20 deq <= 1'b1;    // 10th dequeue (invalid)
87     #20 deq <= 1'b0;
88     #20 $finish;
89     end
90 endmodule

```

FIFO 队列[仿真结果](#)